

Descrierea și modelarea SPICE a dispozitivelor semiconductoare

Dioda și tranzistorul bipolar

Dispozitivele semiconductoare modelate în PSPICE sunt: dioda (D), tranzistorul bipolar (Q), TEC-J (J), TEC-MOS (M) și TEC-GaAs (B).

Dioda

Forma generală de descriere este:


D<nume> <NA> <NC> <nume_model> [aria]

<NA>, <NC> reprezintă nodurile anod (+), respectiv catod (-) ale diodei. Curentul pozitiv prin diodă circulă de la <NA> la <NC>;

<nume_model> reprezintă numele modelului în care sunt descriși parametrii de model ai diodei. Comanda pentru definirea parametrilor de model ai diodei are forma următoare:

.MODEL <nume_model> D [parametri model]

[aria] este un parametru opțional și reprezintă factorul de arie, având valoarea implicită 1. El scalează parametrii de model IS, ISR, IKF, RS, CJO, IBV și este util în special în structurile integrate, unde diodele au aceiași parametri de model, dar pot avea arii diferite. Ca urmare, se definește un singur model pentru toate diodele, urmând să se specifice factorul de arie pentru fiecare în parte.



Exemple:

```
D1 7 23 DMOD
D3 6 2 D1N4001
```

În PSPICE, modelul diodei este unic, atât pentru diodele redresoare, cât și pentru cele de comutație, respectiv stabilizatoare. Caracterizarea fiecărui tip de diodă se face specificând în descrierea modelului doar valorile pentru setul de parametri corespunzători.

Parametrii modelului static al diodei


Modelul diodei constă dintr-o rezistență parazită ($RS / arie$) în serie cu dioda intrinsecă.

Modelul static al diodei are la bază ecuația lui Shockley:

$$I_D = IS \cdot \left[e^{\frac{V_D}{N \cdot V_T}} - 1 \right], \text{ unde } N \text{ este coeficientul de emisie și } IS$$

este curentul de saturație. Această ecuație este valabilă doar la nivele moderate de injecție, iar pentru caracterizarea diodei la nivele mici sau mari de injecție, precum și în conducție inversă, în ecuația lui Shockley se adaugă termeni suplimentari.

În curent continuu, dioda este descrisă de ecuația: $I_D = arie \cdot (I_{direct} - I_{invers})$,



unde $I_{direct} = K_{inj} \cdot IS \cdot \left(e^{\frac{V_D}{N \cdot V_T}} - 1 \right) + K_{gen} \cdot ISR \cdot \left(e^{\frac{V_D}{NR \cdot V_T}} - 1 \right)$, în care K_{inj} reprezintă factorul de injecție mare, K_{gen} reprezintă factorul de generare al curentului de recombinare;

$$I_{invers} = I_{invers\ high} + I_{invers\ low} = IBV \cdot e^{-\frac{V_D + BV}{NBV \cdot V_T}} + IBVL \cdot e^{-\frac{V_D + BV}{NBVL \cdot V_T}}.$$

Pentru un calcul ușor al PSF $g_{min} = 10^{-12}$. Parametrii modelului static al diodei sunt prezentați în tabelul următor.

Parametru model	Semnificație	Valoare implicită	Unitate măsură
IS	Curentul de saturație	1E-14	amp
N	Coeficientul de emisie	1	
ISR	Curentul de recombinare	0	amp
NR	Coeficientul de emisie al curentului de recombinare	2	
IKF	Curentul de cot la nivel mare de injecție	infinit	amp
RS	Rezistența parazită	0	ohm
XTI	Coeficientul de variație cu temperatura a curentului de saturație	3	
EG	Lățimea benzii interzise	1.11	eV
TRS1	Coeficientul de variație liniară cu temperatura a rezistenței parazite	0	°C ⁻¹
TRS2	Coeficientul de variație pătratică cu temperatura a rezistenței parazite	0	°C ⁻²
TIKF	Coeficientul de variație cu temperatura a curentului de cot la nivel mare de injecție	0	°C ⁻¹
BV	Tensiunea de cot la străpungerea inversă	infinit	volt
IBV	Curentul de cot la străpungerea inversă	1E-10	amp
NBV	Factorul de idealitate al străpungerii inverse	1	
IBVL	Curentul de cot al străpungerii inverse, la nivel scăzut	0	amp
NBVL	Factorul de idealitate al străpungerii inverse, la nivel scăzut	1	
TBV1	Coeficientul de variație liniară cu temperatură a tensiunii de cot la străpungerea inversă	0	°C ⁻¹
TBV2	Coeficientul de variație pătratică cu temperatura a tensiunii de cot la străpungerea inversă	0	°C ⁻²

Parametrii modelului de semnal mare al diodei

Modelul de semnal mare al diodei se obține adăugând la parametrii de regim static parametrii ce caracterizează trecerea din starea de blocare în starea de conducție și invers. Trecerea diodei din starea de blocare în conducție și invers are loc cu anumite întârzieri, datorate timpului necesar stocării și eliminării sarcinilor din joncțiune. În joncțiunea p-n există două tipuri de sarcini stocate:

- *Sarcina purtătorilor minoritari*, stocată în regiunile neutre, modelată prin intermediul capacității timpului de tranzit (C_t) :

$$C_t = TT \cdot G_d = TT \cdot \text{aria} \cdot \frac{dI_{\text{direct}}}{dV_D};$$



- *Sarcina spațială a ionilor necompensați* din regiunea de barieră (joncțiune), modelată prin intermediul capacității de barieră (C_j):

$$C_j = \begin{cases} \frac{CJO}{(1 - \frac{V_d}{VJ})^M}, & V_d \leq FC \cdot VJ \\ \frac{CJO}{(1 - FC)^{M+1}} \cdot \left[1 - FC \cdot (1 + M) + M \cdot \frac{V_d}{VJ} \right], & V_d > FC \cdot VJ \end{cases}$$

Reprezentarea grafică a variației capacității de barieră în funcție de tensiunea V_d , de la bornele diodei intrinsece, este prezentată în figura alăturată.

Parametrii de model care intervin în expresiile capacităților C_t și C_j ale diodei și care caracterizează dioda la semnal mare sunt prezentați în tabelul de mai jos.

Capacitatea totală a diodei (C) este:
 $C = C_t + \text{aria} \cdot C_j$.



Parametru model	Semnificație	Valoare implicită	Unitate măsură
CJO	Capacitatea de barieră la polarizare nulă	0	farad
VJ	Potențialul de barieră	1	volt
M	Coeficientul de gradare a joncțiunii	0.5	
TT	Timpul de tranzit	0	sec

Modelul de semnal mic al diodei se obține prin liniarizarea caracteristicii statice în punctul static de funcționare, iar capacitatea totală are valoarea corespunzătoare acestui punct ($V_d = V_D|_{PSF}$):


$$C = C_t|_{PSF} + \text{aria} \cdot C_j|_{PSF}.$$

Zgomotul generat de diodă are două componente, cu următoarele funcții densitate spectrală de putere pe unitatea de bandă:

- zgomot termic generat de rezistența parazită:

$$I_n^2 = \text{aria} \cdot \frac{4kT}{RS};$$

- zgomot de alicie și de licărire generat de dioda intrinsecă:



$$I_n^2 = 2qI_d + KF \cdot \frac{I_d^{AF}}{f}.$$

Parametrii KF și AF sunt alți doi parametri de model ai diodei și au semnificația:
 KF - coeficientul zgomotului de licărire (valoare implicită 0);
 AF - exponentul zgomotului de licărire (valoare implicită 1).


Tranzistorul bipolar

Forma generală de descriere a tranzistorului bipolar este următoarea:

Q<nume> <nod_colector> <nod_bază> <nod_emitor>
 +[nod_substrat] <nume_model> [aria]

Exemple:

Q1 12 8 0 MODNPN
 Q2 9 4 10 MODPNP 1.5
 Q3 7 22 13 0 MODLATPNP 1.2



Forme corespunzătoare descrierii modelului sunt:

.MODEL <nume_model> NPN [parametri model] - pentru tranzistor tip NPN;
 .MODEL <nume_model> PNP [parametri model] - pentru tranzistor tip PNP;
 .MODEL <nume_model> LPNP [parametri model] - pentru tranzistor PNP lateral.

Tranzistorul bipolar este modelat ca un tranzistor intrinsec și rezistențe în serie cu colectorul ($RC / arie$), cu baza ($RB / arie$), cu emitorul ($RE / arie$).

În PSPICE există două modele pentru tranzistoarele bipolare:

- modelul Ebers-Moll, care asigură o precizie scăzută;
- modelul Gummel-Poon, care are la bază modelul Ebers-Moll, dar care are în plus o serie de parametri care modelează mai precis curentul de saturație, capacitățile de tranziție și dependența rezistenței bazei de nivelul de injecție.

Ecuatiile ce descriu **modelul static Ebers-Moll** din figura alăturată sunt:


$$I_C = I_{CT} - \frac{I_{EC}}{BR}; \quad I_E = -I_{CT} - \frac{I_{CC}}{BF}; \quad I_B = \frac{I_{CC}}{BF} + \frac{I_{EC}}{BR},$$

unde:

$$I_{CC} = IS \cdot \left(e^{\frac{qV_{BE}}{kT}} - 1 \right); \quad I_{EC} = IS \cdot \left(e^{\frac{qV_{BC}}{kT}} - 1 \right);$$

$$I_{CT} = I_{CC} - I_{EC}.$$


Parametrii de model corespunzători modelului static Ebers-Moll sunt prezentați în tabelul de mai jos:



Parametru model	Semnificație	Valoare implicită	Unitate măsură
IS	Curentul de saturație	1E-16	amp
BF	Factorul beta maxim ideal în conducție directă	100	
NF	Coeficientul de emisie al curentului în conducție directă	1	
VAF (VA)	Tensiunea Early în conducție directă	infinit	volt
BR	Factorul beta maxim ideal în conducție inversă	1	
NR	Coeficientul de emisie al curentului în conducție inversă	1	
VAR (VB)	Tensiunea Early în conducție inversă	infinit	volt
RE	Rezistența emitorului	0	ohm
RB	Rezistența bazei la PSF nul	0	ohm
RC	Rezistența colectorului	0	ohm

În modelul Ebers-Moll câștigurile direct și invers în curent sunt considerate constante, independente de curentul prin tranzistor, rezistența bazei este considerată constantă, iar caracteristicile de intrare, respectiv de transfer, au pantă constantă față de logaritmul curentului prin tranzistor.

Modelul Gummel-Poon aproximează mult mai bine caracteristicile statice reale ale tranzistorului. În următoarele două figuri de mai jos sunt ilustrate variația câștigului direct în curent în funcție de $\ln(I_C)$, respectiv caracteristicile de transfer ($\ln(I_C)$ în funcție de V_{BE}) și de intrare ($\ln(I_B)$ în funcție de V_{BE}), a căror pantă este variabilă.



Parametrii de model suplimentari ai modelului static Gummel-Poon, față de modelul static Ebers-Moll sunt prezentați în tabelul de mai jos:

Parametru model	Semnificație	Valoare implicită	Unitate măsură
IKF (IK)	Curentul de cot corespunzător câștigului direct în curent la nivel mare de injecție	infinit	amp
ISE (C2)	Curentul de saturație al joncțiunii bază-emitor	0	amp
NE	Coeficientul de emisie al joncțiunii bază-emitor	1.5	
IKR	Curentul de cot corespunzător câștigului invers în curent la nivel mare de injecție	infinit	amp
ISC (C4)	Curentul de saturație al joncțiunii bază-colector	0	amp
NC	Coeficientul de emisie al joncțiunii bază-colector	2	
NK	Coeficient de emisie la curenți mari	0.5	
ISS	Curentul de saturație al joncțiunii substratului	0	amp
NS	Coeficientul de emisie al joncțiunii substratului	1	
RBM	Rezistența minimă a bazei	RB	ohm
IRB	Curentul de bază la care RB este la jumătate între RB și RBM	infinit	amp

Modelul de semnal mare al tranzistorului bipolar

La fel ca la diodă, trecerea tranzistorului bipolar dintr-o stare staționară în alta nu se face instantaneu, ci după un interval de timp. Acest fenomen se datorează celor două tipuri de sarcini stocate în structură:

- sarcinile ionilor ficși din regiunile de tranziție ale joncțiunilor, modelate prin intermediul capacităților de barieră C_{jc} , C_{je} și C_{js} ;
- sarcinile purtătorilor minoritari stocați în regiunile neutre, modelate prin intermediul capacităților timpilor de tranzit direct C_{tbe} , respectiv invers C_{tbc} .

În tabelul de mai jos sunt prezentați parametrii de model care se adaugă la modelul static Ebers-Moll, pentru caracterizarea comportamentului la semnal mare.

Parametru model	Semnificație	Valoare implicită	Unitate măsură
CJE	Capacitatea joncțiunii bază-emitor la polarizare nulă	0	farad
VJE (PE)	Bariera de potențial a joncțiunii bază-emitor	0.75	volt
MJE (ME)	Factorul de gradare al joncțiunii bază-emitor	0.33	
CJC	Capacitatea joncțiunii bază-colector la polarizare nulă	0	farad
VJC (PC)	Bariera de potențial a joncțiunii bază-colector	0.75	volt
MJC (MC)	Factorul de gradare al joncțiunii bază-colector	0.33	
CJS (CCS)	Capacitatea joncțiunii substratului la polarizare nulă	0	farad
VJS (PS)	Bariera de potențial a joncțiunii substratului	0.75	volt
MJS (MS)	Factorul de gradare al joncțiunii substratului	0	
FC	Coeficient de neidealitate al capacității de barieră la polarizare directă	0.5	
TF	Timpul ideal de tranzit direct	0	sec
TR	Timpul ideal de tranzit invers	0	sec

Capacitățile joncțiunilor sunt modelate în PSPICE la fel ca la diodă. Parametrii suplimentari care se adaugă la modelul Gummel-Poon sunt prezentați în tabelul următor.

Parametru model	Semnificație	Valoare implicită	Unitate măsură
XCJC	Fracțiunea din CJC conectată intern la Rb	1	
XTF	Coeficientul dependenței TF de polarizare	0	
VTF	Parametru al dependenței TF de Vbc	infinit	volt
ITF	Parametru al dependenței TF de Ic	0	amp
PTF	Faza în exces la frecvența $1/(2\pi \cdot TF)$ Hz	0	grade

Parametrii de model ce descriu variația cu temperatura a parametrilor tranzistorului sunt prezentați în tabelul următor.

Parametru model	Semnificație	Valoare implicită	Unitate măsură
EG	Banda interzisă	1.11	eV
XTB	Coeficient de variație cu temperatura a BF și BR	0	
XTI (PT)	Exponent variației IS cu temperatura	3	
TRE1	Coeficient liniar de temperatură al RE	0	$^{\circ}C^{-1}$
TRE2	Coeficient pătratic de temperatură al RE	0	$^{\circ}C^{-2}$
TRB1	Coeficient liniar de temperatură al RB	0	$^{\circ}C^{-1}$
TRB2	Coeficient pătratic de temperatură al RB	0	$^{\circ}C^{-2}$
TRM1	Coeficient liniar de temperatură al RBM	0	$^{\circ}C^{-1}$
TRM2	Coeficient pătratic de temperatură al RBM	0	$^{\circ}C^{-2}$
TRC1	Coeficient liniar de temperatură al RC	0	$^{\circ}C^{-1}$
TRC2	Coeficient pătratic de temperatură al RC	0	$^{\circ}C^{-2}$


Efectul de cvasi-saturație

Cvasi-saturația este regiunea de funcționare în care joncțiunea metalurgică bază-colector internă este polarizată direct, în timp ce, la terminalele exterioare, tensiunea de polarizare bază-colector este inversă. Efectul de cvasi-saturație este modelat prin extinderea modelului Gummel-Poon intrinsec. Astfel, sunt adăugate sursa I_{epi} și capacitățile notate Q_w și Q_o din figura de mai jos. Această extindere are loc numai dacă parametrul $RCO > 0$.

Parametrii de model care modelează efectul de cvasi-saturație sunt prezentați în tabelul următor.

Parametru model	Semnificație	Valoare implicită	Unitate măsură
QCO	Factor al sarcinii din regiunea epitaxială	0	coulomb
RCO	Rezistența regiunii epitaxiale	0	ohm
VO	Tensiunea de cot a mobilității purtătorilor	10	volt
GAMMA	Factor de dopare a regiunii epitaxiale	1E-11	

Zgomotul, la fel ca la diodă, are două componente: zgomotul termic, generat de rezistențele parazite RC , RB și RE și zgomotul de alicie și de licărire, generat de curenții de bază și de colector.



Aplicația 1: Caracteristicile statice ale diodei redresoare

```
VIN 1 0 DC 1
D1 1 0 D1N4001
.MODEL D1N4001 D(Is=14.11n N={NVAL} Rs={RSVAL} Ikf=94.81 Xti=3 Eg=1.11
+Cjo=25.89p M=.44 Vj=.3245 Fc=.5 Bv=75 Ibv=10u Tt=5.7u)
.PARAM NVAL=1.984
.PARAM RSVAL=33.89m
.DC VIN 0.4 1.5 5m
.STEP PARAM RSVAL LIST 20m 30m 40m
*.STEP PARAM NVAL LIST 1.5 2 2.5
.PROBE I(D1)
.END
```

Aplicația 2: Caracteristica statică a diodei stabilizatoare

```
VIN 1 0 DC 10
R1 1 2 100
D1 0 2 D02CZ6_8
.LIB NOM.LIB
R2 2 0 RLMOD 1
.MODEL RLMOD RES(R=200)
.DC RES RLMOD(R) 1 2K 1
.PROBE V(2)
.END
```


Aplicația 3: Comportarea diodei redresoare în regim dinamic de comutație

```
VIN 1 0 PULSE(-5 5 10n 10p 10p 30n 60n)
R1 1 2 100
D1 2 0 D1N4001
.LIB NOM.LIB
```

```
*pt.cele 4 combinatii (Cjo=0,Tt=0), (Cjo=0,Tt=5.7u), (Cjo=25.9p,Tt=0),
*(Cjo=25.9p,Tt=5.7u)
.MODEL D1N4001 D(Is=14.11n N=1.984 Rs=33.89m Ikf=94.81 Xti=3 Eg=1.11
+Cjo=25.89p M=.44 Vj=.3245 Fc=.5 Bv=75 Ibv=10u Tt=5.7u)
```

```
.TRAN 1n 100n 0 0.3n
.PROBE V(1) V(2) I(D1)
.END
```

În urma simulării vizualizați și analizați tensiunile din circuit și curentul prin diodă (acesta se va reprezenta într-un plot separat, **Plot-> Add_Plot_to_Window**). Explicați de ce la trecerea din conducție directă în blocare tensiunea pe diodă, pentru un interval de timp, rămâne pozitivă în timp de curentul este negativ!

Aplicația 4: Caracteristicile statice ale tranzistorului bipolar

```
Q1 2 1 0 Q2N2222
.LIB NOM.LIB
IB 0 1 10u
VCE 2 0 5
```

```
*Caract. beta(Ic), caract. de intrare Ib(Vbe) si de transfer Ic(Vbe)
.DC DEC IB 1u 5m 20
```

```
*Caract. de iesire Ic(Vce) pt. Ib=const.
*.DC DEC VCE 1m 10 20
*.STEP IB LIST 100u 200u 300u 400u
```

```
.PROBE
.END
```